
Abstract
We present a higher-order programmer-level technique for
compiling particular kinds of irregular data-parallel prob-
lems to parallel hardware. The technique, which we have
named “flattening-by-expansion” builds on a number of seg-
mented data-parallel operations but is itself implemented as
a higher-order generic function, which makes it useful for
many irregular problems. Concretely, the implementation
is given in Futhark and we demonstrate the usefulness of
the functionality for a number of irregular problems and
show that, in practice, the irregular problems are compiled
to efficient parallel code that can be executed on GPUs. The
technique is useful in any data-parallel language that pro-
vides a key set of primitives.

CCS Concepts • Computing methodologies → Paral-
lel programming languages; • Software and its engi-
neering→ Source code generation; Software performance.

Keywords GPGPU programming, irregular nested paral-
lelism, flattening, functional programming.

1 Introduction
While the development in computer architectures is increas-
ingly providing improved parallel performance characteris-
tics, the world’s programmers have difficulties making effi-
cient use of the increased number of computational parallel

units. For some domains, program abstractions make it pos-
sible for programmers to make use of libraries to release the
potential of new hardware, while such an approach can be dif-
ficult to apply for other domainswhere high performance can
be achieved only by parallelising particular domain-specific
algorithms. Ideally, one could hope that existing code for
such algorithms could be compiled unchanged to exploit
the unreleased power of the new architectures. However,
although such an approach works sometimes, in general,
it does not. In such cases, programmers will often need to
master parallel programming using low-level abstractions,
such as those provided by CUDA and OpenCL, for which
the learning curve is steep and the maintenance and devel-
opment costs are high.
Modern hardware favors regular data-parallel patterns

and it is well understood how also nested regular patterns
can be transformed into flat regular parallelism. However,
nested irregular parallelism introduces problems that are not
easily dealt with. In particular, the overhead of managing
segment descriptors (i.e., data that describes the irregularity)
and the additional overhead of applying segmented opera-
tions become problematic, and, often, the overhead becomes
difficult for programmers to understand and reason about.

In 1990, Blelloch introduced the functional programming
language NESL [5], a high-level first-order parallel functional
language, centered around the idea that nested parallelism
(and also irregular parallelism) could be eliminated at com-
pile time by a transformation called flattening. NESL was
built around a single parallel array comprehension construct,
which, as it turned out, could be used to implement a se-
ries of parallel constructs such as map, reduce, filter, and scan

[4, 6–8]. This flattening approach to supporting irregular
parallelism has later been refined for efficiency purposes
[2, 10, 11, 24]. However, in general, it is difficult for the gen-
eral flattening techniques to compete with hand-optimised
flattened code.

Futhark is a statically typed parallel functional array lan-
guage, which supports well a notion of so-called moderate
flattening, which allows for many cases of regular nested
parallelism to be mapped to efficient flat parallelism [19].
Whereas regularly nested parallel map constructs can be
translated trivially to flat parallelism, it is not immediately
clear, in general, how to support efficiently non-regular
nested map constructs. Futhark further implements a no-
tion of incremental flattening [20], which generates multiple

code versions in situations where the most optimal flatten-
ing strategy depends on the properties of the input data.
For instance, the most optimal GPU code for dense matrix-
multiplication depends highly on the sizes of the matrix
dimensions. For dealing with irregular problems, Futhark
requires the programmer to implement the flattening strat-
egy by hand, which can be quite cumbersome. However, the
programmer may choose different strategies for the imple-
mentation, which in some cases could involve padding (not
work efficient but sometimes the most performance efficient
technique in practice) or full flattening, which leads to work
efficient implementations, which, however, are sometimes
slow in practice.1

In this paper, we present a design pattern for obtaining a
full-flattened implementation of a certain class of irregular
data-parallel problems. The design pattern is implemented
in Futhark as a generic higher-order function expand, which
has the following generic type:

val expand 'a 'b : (a → i32) → (a → i32 → b)

→ []a → []b

The function expands a source array, of type []a, into a tar-
get array, of type []b, given (1) a function that determines,
for each source element, how many target elements it ex-
pands to and (2) a function that computes a particular tar-
get element based on a source element and the target el-
ement index associated with the source. As a simple ex-
ample, the expression expand (\x→x) (*) [2,3,1] returns the
array [0,2,0,3,6,0]. Here (\x→x) denotes the identity func-
tion and (*) denotes integer multiplication. Semantically,
expand f g xs performs the equivalent of

flatten (map (\x → map (g x) (iota (f x))) xs)

where iota n produces the array of integers from 0 up to
n-1, and flatten flattens an array of dimension n + 1 into an
array of dimension n. Notice that the inner map operates on
an array of f x elements, which is variant to the outermost
map. Thus, the array passed to flatten is potentially irregular.
The purpose of expand is to support this kind of irregular
map nests in languages that do not directly support irregu-
lar parallelism. Given the usual definition of a fused imple-
mentation of flatten and map, called flatMap, which has type
(a → []b) → []a → []b, the semantics of expand can also be
given by the equation

expand f g = flatMap (\x → map (g x) (iota (f x)))

As an example of using the expand function to solve an
irregular problem, consider the task of finding the points in
1A parallel algorithm is said to be work efficient if the work (i.e., number
of operations) performed by the algorithm is of the same asymptotic com-
plexity as the work performed by the best known sequential algorithm that
solves the same problem.

Figure 1. A grid of lines. Each line is defined by its end
points and the number of points that make up a line can be
determined based on the maximum of the distances between
the x coordinates and the y coordinates of the end points.

a 2d plane that constitute an array of line segments, each
given by its end points; see Figure 1 for an example of a grid of
lines. The technique wewill use to “draw lines” resembles the
development by Blelloch [5] with the difference that it makes
use of the expand function and that the underlying language
implementation does not support irregular parallelism. Using
the expand function, all we need is to provide (1) a function
that determines for a given line, the number of points that
make up the line and (2) a function that determines the n’th
point of a particular line, given the index n. The code for
such an approach is listed in Figure 2.

The function points_in_line makes use of the observation
that the number of points that make up the constituting set
of points, for the line with end points (x1,y1) and (x2,y2),
is one plus the maximum of |x2 − x1 | and |y2 − y1 |, that
is, one plus the maximum of the absolute values of the dif-
ference in x-coordinates and y-coordinates, respectively.2
Using this observation, the function get_point_in_line can
independently compute the i’th point in the line by first
calculating the proper direction and slope of the line (the
two utility functions), relative to the line’s starting point. A
conditional expression guides whether the x-dimension or
the y-dimension is dominating.
Using the flattening-by-expansion approach, we obtain

a work efficient implementation of line drawing with work
and span complexity being bounded by the work and span
complexity of the underlying complexity properties of scan,
which the implementation of expand is founded on.3 In the
concrete case, the two function arguments passed to expand

are constant time operations, which means that, with n being
the number of resulting points, the work complexity of the

2In Futhark, the dot-notation is overloaded and used for tuple- and record-
projection, module access, and for locally opening a module inside the
parentheses following the dot.
3By work we refer to the total number of operations performed and by span
we refer to the length of the longest chain of dependent parallel operations
(also sometimes called depth.

type point = (i32 ,i32)

type line = (point ,point)

let points_in_line ((x1,y1),(x2,y2)) =

i32 .(1 + max (abs(x2-x1)) (abs(y2-y1)))

let compare (v1:i32) (v2:i32) : i32 =

if v2 >v1 then 1 else if v1>v2 then -1 else 0

let slope (x1 ,y1) (x2,y2) : f32 =

if x2==x1 then if y2>y1 then 1 else -1

else r32(y2 -y1) / f32.abs(r32(x2-x1))

let get_point_in_line ((p1,p2):line) (i:i32) =

if i32.abs(p1.1-p2.1) > i32.abs(p1.2-p2.2)

then let dir = compare p1.1 p2.1

let sl = slope p1 p2

in (p1.1+dir*i,

p1.2+t32(f32.round(sl*r32 i)))

else let dir = compare p1.2 p2.2

let sl = slope (p1.2,p1.1) (p2.2,p2.1)

in (p1.1+t32(f32.round(sl*r32 i)),

p1.2+i*dir)

let points_of_lines (ls: []line) : []point =

expand points_in_line get_point_in_line ls

Figure 2. Code for drawing lines in parallel. We make use
of the truncation function t32 : f32 -> i32.

algorithm is O (n) and the span complexity of the algorithm
is O (logn).
The contributions of this paper are the following:

1. We present a generic approach to implementing a class
of irregular (and even nested) data-parallel problems
using a language-agnostic construct called expand.

2. We present an implementation of expand in Futhark,
based on low-level segmented operations.

3. We demonstrate the usefulness of the flattening-by-
expansion technique by showing that it can be used
for flattening a variety of real-world problems.

4. We demonstrate that the implementation of expand

leads to efficient implementations of problems in prac-
tice by comparing the performance of the implemen-
tations with hand-flattened code in some cases.

5. We discuss the limitations of the approach and give an
example of an extension of expand, called expand_reduce,
which can be used for implementing, for instance,
sparse matrix-vector multiplication.

The remainder of the paper is organised as follows. In the
following sections, we present the details of the implemen-
tation of the expand function and the underlying segmented
operations that the implementation builds on. In Section 4,

we demonstrate how the expand function can be used for im-
plementing nested irregular parallelism. In particular, we
show how it can be used to expand an array of triangles
or circles into an array of lines, which can then be further
expanded into an array of points. In Section 5, we show how
we can implement a work-efficient data-parallel implemen-
tation of Eratosthenes’ sieve and in Section 6, we show how
the approach can be used to implement sparse matrix-vector
multiplication. In Section 7, we show how the technique can
be combined with some of Futhark’s more elaborate reduc-
tion constructs for controlling the depth of objects when
drawing graphics. In Section 8, we describe related work and
in Section 9, we describe future work and conclude.

All code for the presented examples are available at https:
//github.com/diku-dk/futhark-array19.

2 A Toolbox of Segmented Operations
Futhark features a number of low-level data-parallel con-
structs, including map, reduce, scan, and filter. Futhark is a
sequentialising compiler in that in case some of the parallel
constructs may turn up inside already parallel constructs,
the compiler is permitted to sequentialise them if it judges
this will result in the most efficient code. Futhark imple-
ments a number of fusion and flattening transformations
that will seek to get as good a performance as possible while
maintaining the semantics of the program.
Futhark also features good abstraction mechanisms in-

cluding higher-order functions [21], polymorphism, record
types, and higher-order modules [14], which are all features
that are present in the source language, but eliminated at
compile time with the aim of obtaining efficient target code.

2.1 Segmented Scan
A key operation needed for working with irregular prob-
lems is a segmented scan operation. Whereas specialised
segmented scan implementations exist, in Futhark, a seg-
mented scan operation can be defined using the ordinary
scan function. Following Blelloch [5, Section 13.2], a generic
segmented scan operation can be implemented as follows:

let segm_scan [n] 't (op: t → t → t) (ne: t)

(flags: [n]bool)

(as: [n]t) : [n]t =

zip flags as

|> scan (\(x_flag ,x) (y_flag ,y) →

(x_flag || y_flag ,

if y_flag then y else x `op` y))

(false , ne)

|> unzip |> (.2)

The first to notice about the Futhark implementation of
segm_scan is that it is parametric in the type of elements and
that Futhark also allows for specifying, using a so-called
size-parameter, that the two array arguments should have

the same size (i.e., n) and that the resulting array also has size
n. This simple support for certain kinds of dependent typing
helps the compiler eliminate a number of dynamic checks
while at the same time allowing the programmer to specify
simple contractual properties of functions. As we shall see
later, size parameters may also be referenced as ordinary
variables of type i32 in the body of the function, which often
makes it straightforward to refer to the size of an argument
array.
Given a binary associative operator op with neutral el-

ement ne, the function computes the inclusive prefix scan
of the segments of as specified by the flags array, where
true starts a segment and false continues a segment. It is a
straightforward exercise to prove that, given op is an associa-
tive operator with neutral element ne, the function argument
passed to scan is an associative operator and (false,ne) is its
neutral element.

Futhark implements arrays of records (or tuples) as records
(or tuples) of arrays, which means that source language zip

and unzip operations are compiled into the identity function,
which has zero overhead. The higher-order infix “pipe opera-
tor” |> passes the result of its left-hand side into the function
to the right.

2.2 Replicated Iota
Based on the segm_scan function, we will now present an
important utility function called repl_iota. Given an array
of natural numbers, specifying repetitions, the function re-
turns an array of weakly increasing indices (starting from 0)
and with each index repeated according to the entry in the
repetition array. As an example, repl_iota [2,3,1,1] returns
the array [0,0,1,1,1,2,3]. The function is defined in terms
of other parallel operations, including scan, map, iota, scatter,
replicate, reduce, and, as mentioned, segm_scan.
Futhark’s scatter function is specified as follows:

val scatter 't : *[]t → []i32 → []t → *[]t

The first array argument is modified inplace with an associ-
ation list of updates specified by the following two arrays.
The modified array is then transferred back to the caller
of scatter. Notice that the uniqueness typing (the *’s in the
type), functions here as a simple ownership transfer mecha-
nism.

Here is the definition of repl_iota:

let repl_iota [n] (reps:[n]i32) : []i32 =

let s1 = scan (+) 0 reps

let s2 = map (\i → if i==0 then 0

else unsafe s1[i-1]) (iota n)

let tmp =

scatter (replicate (reduce (+) 0 reps) 0)

s2 (iota n)

let flags = map (>0) tmp

in segm_scan (+) 0 flags tmp

Whereas the binding of s1 results in an inclusive scan of the
repetition values, the binding of s2 results in an exclusive
scan. Using the tmp array, which will be of size equal to the
resulting array, the flags array will contain true values in
positions where the indexes should be increased (and zeros
elsewhere). The final segmented scan operation will return
the desired result.

Notice that in order to use this Futhark code with futhark
opencl, we need to prefix the array indexing in line 4 with
the unsafe keyword; the reason is that Futhark is not suffi-
ciently smart to convince itself that the array indexing in
line 4 is always within bounds.
An example evaluation of a call to the function repl_iota

is provided in the following table:

Arg/result
reps = [ 2 3 1 1 ]
s1 = [ 2 5 6 7 ]
s2 = [ 0 2 5 6 ]
replicate = [ 0 0 0 0 0 0 0 ]
tmp = [ 0 0 1 0 0 2 3 ]
flags = [ 0 0 1 0 0 1 1 ]
segm_scan = [ 0 0 1 1 1 2 3 ]

The resulting array is shown in the last line.

2.3 Segmented Iota
Another useful utility function is the function segm_iota,
which, when given an array of flags (i.e., booleans), returns
an array of catenated index sequences, each of which is re-
set according to the booleans in the array of flags. As an
example, the expression

segm_iota [false ,false ,false ,true ,false ,false]

returns the array [0,1,2,0,1,2]. The segm_iota function can
be implemented with the use of a simple call to segm_scan

followed by a call to map:

let segm_iota [n] (flags:[n]bool) : [n]i32 =

segm_scan (+) 0 flags (replicate n 1)

|> map (\x → x-1)

The map function call is necessary because segm_scan imple-
ments a segmented inclusive scan (contrary to a segmented
exclusive scan for which each segment is initiated with an
occurrence of the neutral element). Notice that the size-
parameter n helps specifying that the size of the result array
is of the same size as the given flags array.

3 The Expand Function
Using the utility functions defined in the previous section, we
can now define the expand function. The function is listed in
Figure 3. The function makes use of the two utility functions

let expand 'a 'b (sz: a → i32) (get: a → i32 → b)

(arr:[]a) : []b =

let szs = map sz arr

let idxs = repl_iota szs

let iotas = segm_iota (map2 (!=) idxs

(rotate (-1) idxs))

in map2 (\i j → get (unsafe arr[i]) j)

idxs iotas

Figure 3. The definition of the expand function.

repl_iota and segm_iota, which were both presented in Sec-
tion 2. Assuming that sz and get are constant functions, the
dominating function calls of expand are the segmented scan

operations appearing inside repl_iota and segm_iota. These
calls operate on data of the size M =

∑
x∈a sz x, where a

is the array argument passed to expand. Under the assump-
tions of sz and get being constant functions, the work and
span complexity of expand is therefore O (M ) and O (logM ),
respectively.

The expand function can be defined in any parallel language
that provides a suitable small set of primitives, namely map,
segmented prefix sum (or simply a scan that supports an
arbitrary operator, as in Futhark), scatter, and gather. Notice
that support for nested parallelism is not required.

3.1 Algebraic Properties and Fusion
The introduced expand function features a number of alge-
braic properties. We have already presented the semantics
of expand in terms of iota, map, and flatten (and an alternative
specification in terms of flatMap, which is also sometimes
called concatMap).

Another simple algebraic property, which can be used to
convert a program into using expand (if that is desired), is the
following:

expand (const 1) f = map (\x → f x 0)

Regular uses of expand can be converted into a regular map-
nest using the following algebraic property:

expand (const n) f xs =

flatten <-< map (\i → map (\x → f x i) xs)

(iota n)

Here the infix Futhark function <-< denotes function compo-
sition. Futhark does not currently recognise such patterns
as expand is a user defined function. Futhark will, however,
happily inline the const function inside the expand function
at every use of expand. This inlining could potentially give
rise to optimisations, which, however, are not currently ex-
ploited.

A proper fusion scheme is essential for any language that
targets GPUs [10, 16, 25, 29]. Futhark implements a num-
ber of fusion strategies but is also careful not to introduce
duplication of work [18].

The expand function fuses with map and filter as follows:

map f <-< expand sz get = -- map

expand sz (\x → f <-< get x) -- fusion

expand sz get <-< filter p = -- filter

expand (\x → if p x then sz x -- fusion

else 0) get

Because Futhark supports well map-map fusion [18] and
because applications of expand are inlined by Futhark, essen-
tially, expand fuses with map. Fusing expand with filter, how-
ever, is not easily supported, however, unless the Futhark
fusion engine gets to learn about the intrinsic fusion proper-
ties of expand.

4 Nested Irregular Parallelism
In this section, we demonstrate that the flattening-by-expansion
technique can also be applied in a nested setting with flat-
tening happening at multiple levels.

4.1 Drawing Triangles
An example of an algorithm worthy of flattening is triangle
rasterisation, that is, an algorithm that in parallel computes
the points that constitute a set of triangles. The algorithm
that we present here is based on the assumption that we al-
ready have a function for drawing multiple horizontal lines
in parallel. Luckily, we have already seen how we can de-
fine such a function! The algorithm for drawing triangles is
based on the property that any triangle can be split into an
upper triangle with a horizontal baseline and a lower triangle
with a horizontal ceiling. Just as the algorithm for drawing
lines makes use of the expand function defined earlier, so will
the flattened algorithm for drawing triangles. A triangle is
defined by the three points representing the corners of the
triangle:

type triangle = (point , point , point)

We shall make the assumption that the three points that
define the triangle have already been sorted according to
the y-axis. Thus, we can assume that the first point is the
top point, the third point is the lowest point, and the second
point is the middle point (according to the y-axis).
The first function we need to pass to the expand function

is a function that determines the number of horizontal lines
in the triangle:

let lines_in_triangle ((p,_,r): triangle) : i32 =

r.2 - p.2 + 1

The second function we need to pass to the expand function
is somewhat more involved. We first define a function dxdy,
which computes the inverse slope of a line between two
points:4

let dxdy (a:point) (b:point) : f32 =

let dx = b.1 - a.1

let dy = b.2 - a.2

in if dy == 0 then 0

else r32 dx / r32 dy

We can now define the function that, given a triangle and
the horizontal line number in the triangle (counted from the
top), returns the corresponding line:

let get_line_in_triangle ((p,q,r): triangle)

(i:i32) =

let y = p.2 + i in

if i <= q.2 - p.2 then -- upper half

let sl1 = dxdy p q

let sl2 = dxdy p r

let x1 = p.1+t32(f32.round(sl1*r32 i))

let x2 = p.1+t32(f32.round(sl2*r32 i))

in ((x1 ,y),(x2,y))

else -- lower half

let sl1 = dxdy r p

let sl2 = dxdy r q

let dy = (r.2 - p.2) - i

let x1 = r.1-t32(f32.round(sl1*r32 dy))

let x2 = r.1-t32(f32.round(sl2*r32 dy))

in ((x1 ,y),(x2,y))

The function distinguishes between whether the line to
compute points for resides in the upper or the lower subtrian-
gle. Finally, we can define a parallel, work-efficient function
that converts a number of triangles into lines:

let lines_of_triangles (xs:[] triangle) : []line =

expand lines_in_triangle get_line_in_triangle

(map normalise xs)

In the above function, the function normalize sorts (using
an unrolled bubble sort) the corner points in each triangle
according to the y-axis:

let normalise ((p,q,r): triangle) : triangle =

let bubble (a:point) (b:point) =

if b.2 < a.2 then (b,a) else (a,b)

let (p,q) = bubble p q

let (q,r) = bubble q r

let (p,q) = bubble p q

in (p,q,r)

4For converting floats to integers, we make use of the function r32 : i32
-> f32.

Figure 4. A grid of points generated by first, in parallel,
generating the lines that make up a number of triangles, and
then, also in parallel, generating the points that make up the
lines. The entire algorithm is work-efficient due to flattening
and the use of the expand function.

Figure 4 shows the code in action when the function
lines_of_triangles is called with an array of three triangles,
defined as follows:

[((5,10), (2,28) , (18,20)),

((42,6), (58,10), (25,22)),

((8,3) , (15,15), (35 ,7))]

The lines generated by the function lines_of_triangles is fur-
ther processed using the points_of_lines function, which gen-
erates the points that are then shown in a grid of height 30
and width 62.
The technique demonstrated for triangles can easily be

adapted to work for solid circles and ellipses. The technique
can also be adapted to work for drawing the circumference
of regular polygons and circles.

5 Flattening the Sieve of Eratosthenes
A sometimes useful strategy for obtaining a parallel algo-
rithm is to use the concept of contraction, the general algo-
rithmic trick of solving a particular problem by first making
a contraction step, which simplifies the problem size, and
then repeating the contraction algorithm until a final result
is reached [27]. A variant of a contraction algorithm is an
algorithm that first solves a smaller problem, recursively,
and then uses this result to provide a solution to the larger
problem. One such algorithm is a version of the Sieve of
Eratosthenes that, to find the primes smaller than some n,
first calculates the primes smaller than

√
n. It then uses this

intermediate result for sieving away the integers in the range
√
n up to n that are multiples of the primes smaller than

√
n.

Unfortunately, Futhark does not presently support recur-
sion, thus, one needs to use a loop construct instead to im-
plement the sieve. A Futhark program calculating an array
containing the set of primes below some number n, is shown
in Figure 5.

let primes (n:i32) : []i32 =

(.1) <|

loop (acc ,c) = ([],2) while c < n+1 do

let c2 = if c < t32(f32.sqrt(r32(n+1)))

then c*c

else n+1

let is = map (+c) (iota(c2-c))

let fs = map (\i →

let xs = map (\p → if i%p==0

then 1

else 0) acc

in reduce (+) 0 xs) is

-- apply the sieve

let new = filter (\i → 0 == unsafe fs[i-c]) is

in (acc ++ new , c2)

Figure 5.Non-flattened version of the Sieve of Eratosthenes.

Notice that, when computing c2, we are careful not to
introduce overflow by not calculating c*c unless c is less than
the square root of n+1. Notice also that the algorithm applies a
parallel sieve for each step, using a combination of maps and
reductions. The best known sequential algorithm for finding
the number of primes below somen takes timeO (n log logn).
Although the present algorithm is quite efficient in practice,
it is not work efficient, since the work inside the loop is super-
linear. The loop itself introduces a span with a log logn factor
because the size of the problem is squared at each step, which
is identical to doubling the exponent size at each step (i.e.,
the sequence 22, 24, 28, 216, . . . ,n, where n = 22m , for some
positivem, hasm = log logn elements.)

We now present a work-efficient variant of the above func-
tion for computing primes using the concept of flattening-
by-expansion. The function is listed in Figure 6.
There are two points to notice about the code. First, we

use the expand function to calculate and flatten the sieves.
For each prime p and upper limit c2, we can compute the
number of contributions in the sieve (the function sz). Then,
for each prime p and sieve index i, we can compute the sieve
contribution (the function get). Second, using a scatter, a map,
and a filter, we can compute the new primes in the inter-
val c to c2. The algorithm presented here is work efficient,
has work complexity O (n log log n), and span complexity
O (log log n).

We have compared the two Futhark algorithms running
with the OpenCL backend to a straightforward C implemen-
tation of the Sieve of Eratosthenes for finding the number of
primes below 100,000,000. The experiments were performed
on a machine with an NVIDIA RTX 2080 Ti GPU (using
futhark opencl) and an Intel Xeon E5-2650 running at
2.60GHz (using futhark c). We report timing results as
averages over 10 runs. The plain C version takes 530ms on
average, the non-flattened version takes 171ms on average,

let primes (n:i32) =

(.1) <|

loop (acc:[]i32 ,c) = ([],2) while c < n+1 do

let c2 = if c < t32(f32.sqrt(r32(n+1)))

then c*c

else n+1

let sz (p:i32) = (c2 - p) / p

let get p i = (2+i)*p

let sieves : []i32 = map (\p → p-c)

(expand sz get acc)

let vs = replicate (c2-c) 1

let vs = scatter vs sieves

(replicate (length sieves) 0)

let new = filter (>0) <| map2 (*) vs (c..<c2)

in (acc ++ new , c2)

Figure 6. Flattened version of the Sieve of Eratosthenes
using flattening-by-expansion.

and the flattened version, which uses the expand function,
takes 11.3ms on average. We emphasise here that we have
arranged that the versions are comparable in the sense that
they all compute the sieves from scratch.
For these and later measurements, we do not measure

the time taken to move input data to the GPU, or results
back to the CPU. The operations in this paper tend to be
building blocks in larger Futhark programs, not full appli-
cations, and it is our experience that in practice, their data
is already located on the GPU, and their results also need
further processing on the GPU.

6 Sparse Matrix-Vector Multiplication
Numerous possible representations of sparse matrices exist.
Here we demonstrate the use of the flattening-by-expansion
technique for implementing a version of sparse-matrix vec-
tor multiplication based on a compressed sparse row im-
plementation of sparse matrices. In Futhark, the type of a
compressed sparse row representation of a matrix can be
defined as follows:

type csr 't = {row_off: []i32 ,

col_idx: []i32 ,

vals: []t}

The type csr is parameterised over the type of the underlying
matrix values. Given a sparse matrix of size N ×M with NNZ
non-zero values, the size of the row_off array is N + 1 and
the size of each of the col_idx and vals arrays is NNZ. The
compressed sparse row representation favours that each row
can be processed in parallel. However, because each row con-
tains a different number of non-zero elements, the problem
becomes irregular. We shall apply an extended version of
the expand function, which has the following type:

let smvm ({row_off ,col_idx ,vals} : csr f32)

(v:[] f32) : []f32 =

let rows = map (\i → (i,

row_off[i],

row_off[i+1]- row_off[i]))

(iota(length row_off - 1))

let sz r = r.3

let get r i = vals[r.2+i] * v[col_idx[r.2+i]]

in expand_reduce sz get (+) 0f32 rows

Figure 7. Flattened implementation of sparse matrix-vector
multiplication in Futhark.

val expand_reduce 'a 'b [n] : (a → i32)

→ (a → i32 → b)

→ (b → b → b) → b → [n]a → [n]b

The expand_reduce function makes use of the internal flags
array (defined inside expand) to call a function segm_reduce

with the arguments of type (b → b → b) and b to reduce the
expanded values. The function segm_reduce can be defined in
terms of the segm_scan operation, but we shall not give the
definition here.

Using the expand_reduce function, sparsematrix-vectormul-
tiplication can be defined as shown in Figure 7. The function
smvm expands each row into an irregular number of multiplica-
tions and then reduces these multiplication results using the
monoid defined by the associative function (+) and its neutral
element 0. Letting smvm be parameterised over multiplication
and the monoid structure would allow us to instantiate the
smvm function to work on, for instance, matrices and vectors
containing f64 values.

The following table shows the performance of multiplying
different sparse matrices of dimension 10, 000 × 10, 000 with
a dense vector of size 10, 000. Again, the experiments were
performed on a machine with an NVIDIA RTX 2080 Ti GPU
(using futhark opencl) and an Intel Xeon E5-2650 running
at 2.60GHz (using futhark c). We report timing results
(averages over 10 runs) for densities of 1-20 percent:

Density
Futhark
C (ms)

Futhark
OpenCL (ms)

1% 27.1 0.52
2% 49.9 0.79
3% 74.1 1.10
4% 97.0 1.40
5% 130.3 1.70
10% 264.6 3.12
15% 381.2 4.55
20% 506.6 5.90

Dense 94.01 2.77
The sparse matrix density (reported in percentages) corre-
sponds to the number of non-zero elements (in millions). The

last entry shows numbers for a dense matrix-vector multi-
plication, and we see that when the density gets higher than
approximately 10 percent, dense matrix-vector multiplica-
tion outperforms the sparse version. This is because of lower
constant factors; for the dense matrix-vector multiplication,
the Futhark compiler generates a transposition to ensure
coalesced memory access, followed by a call to a single GPU
kernel that performs the actual computation. In contrast, the
sparse operation requires several expensive scans.

6.1 Sparse Matrix-Matrix Multiplication
It turns out that it is straightforward to implement sparse
matrix-matrix multiplication on top of the functionality al-
ready developed. Here is a function that implements multi-
plication of a sparse matrix with a dense matrix:

let smmm [n] (sm:csr f32) (m:[][n]f32) :[n][]f32 =

map (smvm sm) (transpose m)

It is more difficult to implement matrix multiplication
between two sparse matrices, for which efficient implemen-
tations require some degree of binary searching.

7 Managing Graphics Depth
The triangle drawing technique presented in Section 4.1
works only when all triangles have the same color. In essence
when two triangles overlap, the lines, and eventually the
points, generated with lines_of_triangles and points_of_lines

can be used in concert with scatter to draw the triangles on
a canvas. However, scatter does not provide any guarantees
about the effect of writing multiple values to the same entry,
except when the values are identical.
To deal with this problem, Futhark features a function

called reduce_by_index, which can be used instead of scatter
to control the effect of multiple writes to the same entry.
Here is the type of the function:

val reduce_by_index 'a : *[]a → (a → a → a) → a

→ []i32 → []a → *[]a

In addition to the array arguments, also taken by scatter,
reduce_by_index also takes an (assumed to be) associative and
commutative function, operating on array elements, and its
neutral element. This function is used to combine the old
value in the array with the new one that is being written.
We shall not discuss the implementation of reduce_by_index
here but just mention that its implementation is based on the
techniques used for implementing histogram computations
on GPUs [23, 26]. In fact, reduce_by_index can be viewed as a
generalised function for computing histograms.
Using a 3d representation of colored points, lines, and

triangles, we can now make use of reduce_by_index to control
which parts of triangles are shown. We do this by pairing
each pixel to be written with its distance from the camera,

Figure 8. A landscape scene from a game implemented in
Futhark. The terrain is displayed using a large number of
triangles, which are colored according to their vertical posi-
tions in the terrain.

and providing an operator to reduce_by_index that picks the
pixel closest to the camera. This technique is exactly the
classic technique of z-buffering.
Figure 8 shows a scene from a game implemented in

Futhark, where the landscape is drawn using a set of tri-
angles, colored differently based on the vertical position in
the terrain.

Using the AMD Radeon Pro 460 GPU on a MacBook Pro,5
500,000 triangles can be drawn using flattening-by-expansion
with a frame rate of 15 frames per second (on a 2880 × 1800
display). These numbers include the time for computing
the triangles, based on the camera’s point-of-view and the
terrain information, and for copying the computed images
back and forth between the GPU and CPU (Futhark cannot
presently store images directly in image buffers). Notice that
this implementation is of course much less efficient than
if we used the specialised graphics hardware on the GPU.
We do not claim competitive rendering performance; merely
that expand allows us to express the algorithm in a natural
and parallel way, and still obtain decent performance.

8 Related Work
Much related work has been carried out in the area of sup-
porting nested parallelism, including the seminal work on
flattening of nested parallelism in NESL [5, 6], which was
extended to operate on a richer set of values in Data-parallel
Haskell [9], and the work on data-only flattening [34]. These
approaches tend to focus on maximising expressed paral-
lelism, and negate the need for a function such as expand.
However, compiler-based flattening has proven challenging
to implement efficiently in practice, particularly on GPUs [3].
Other promising attempts at compiling NESL to GPUs in-
clude Nessie [28], which is still under development, and
CuNesl [34], which aims at mapping different levels of nested
parallelism to different levels of parallelism on the GPU, but
lacks critical optimisations such as fusion.

5In contrast to the compute servers we are using for the other benchmarks,
a MacBook Pro has a screen.

More recent data-parallel languages include Obsidian [12,
30, 31] and Accelerate [10], which are both embedded in
Haskell, and do not feature arbitrary nested parallelism. Ac-
celerate in particular can easily support manually flattened
programming in the expand style, as segmented scans and
scatter operations are readily available. Accelerate also sup-
ports certain forms of irregular arrays by supporting a notion
of irregular stream scheduling [13].
Other attempts at supporting nested (and even irregular)

parallelism on GPUs include more dynamic approaches, such
as dynamic thread block launching [33] and dynamic par-
allelism, which are extensions to the GPU execution model
involving runtime and micro architecture changes. These
approaches to supporting irregular parallelism does, how-
ever, often come with a significant overhead [32]. Other
dynamic approaches include a partial flattening approach,
implemented using thread stealing, which also introduce a
significant overhead [22].

9 Conclusions and Future Work
In this paper, we have demonstrated a programming tech-
nique that allows for convenient manual flattening of certain
irregular nested parallel constructs, even if the target lan-
guage does not support nested parallelism at all. The result-
ing code is asymptotically as efficient as that which would
have been generated with full NESL-style flattening, and
allows the programmer more control and a “pay-as-you-go
strategy” to flattening. Further, the real-world performance
is sufficient to carry out real-time graphics rendering.

There are a number of possibilities for future work. First,
some overhead can perhaps be avoided in situations where
flattened data is immediately scattered into a target array.
To avoid the resulting double copying overhead, one may
consider defining a function that instead of returning a target
array takes as argument a destination array, which is then
returned to the caller with modified content. Second, there
are a number of irregular nested parallel algorithms that may
benefit from the use of the expand function. Such algorithms
include algorithms for graph traversals [17] and irregular
parallel financial applications [1].

Other possible future work include investigating whether
the technique can be extended in such a way that it can be
used to ease the flattening of more involved algorithms, such
as quick-sort [15] or multiplication of two sparse matrices.

